CN # 1,2 / 2 marzo 2015 / C43

- Pagina web del corso ...
- Programma ...

    (+) ARITMETICA del CALCOLATORE

    1) ZERI di funzioni (es: $f(x) = 0$ ...)

    2) SISTEMI di EQUAZIONI LINEARI (es: $Ax = b$ ...)

        * metodi DIRETTI (fattorizz di A ...)
        * metodi ITERATIVI (matrici SPARSE)

    3) EQUAZIONI DIFFERENZIALI ORDINARIE (es: $\ddot{x} = F(\dot{x}, x)$ ...)

$\boxed{1}$ ZERI di FUNZIONI & ARITMETICA del CALCOLATORE

Pb: data $f: [a,b] \to \mathbb{R}$ continua t.c.

$\exists\, \alpha \in \mathbb{R}$ t.c. $f(\alpha) = 0$, determinare α.

"ZERO di f"

> TEO (esistenza degli zeri)
>
> $f: [a,b] \to \mathbb{R}$ continua t.c. $f(a)\,f(b) < 0$
>
> $\Rightarrow \exists\, \alpha \in (a,b)$ t.c. $f(\alpha) = 0$

- Metodo di BISEZIONE

Idea: utilizz il Teo di esistenza degli zeri per ottenere una succes di intervalli $I_k = [a_k, b_k]$ t.c.

- $\forall k$, $\exists$ zero di f in I_k
- $I_{k+1} \subset I_k$
- $\lim_{k \to \infty} \text{mis } I_k = 0$

descrizione del METODO di BISEZIONE

dati: $f: [a,b] \to \mathbb{R}$ t.c. $f(a) f(b) < 0$

- $a_0 = a$; $b_0 = b$; $I_0 = [a_0, b_0]$; $x_0 = \frac{a_0 + b_0}{2}$;
- per $k = 1, 2, 3, \ldots$ ripeti:

  se $f(x_{k-1}) = 0$ allora: STOP, altrimenti

  - se $f(x_{k-1}) f(b_{k-1}) < 0$ allora: $a_k = x_{k-1}$, $b_k = b_{k-1}$;
  - se $f(a_{k-1}) f(x_{k-1}) < 0$ allora: $a_k = a_{k-1}$, $b_k = x_{k-1}$;
  - $I_k = [a_k, b_k]$; $x_k = \frac{a_k + b_k}{2}$;

uscita: quando un opportuno criterio d'arresto è verificato: x_k (pto medio dell'ultimo intervallo determinato)

Oss: • $\text{mis } I_k = b_k - a_k = \frac{\text{mis } I_{k-1}}{2^1} = \frac{\text{mis } I_{k-2}}{2^2} = \ldots$

$$\ldots = \frac{\text{mis } I_0}{2^k} \Rightarrow \boxed{\lim_{k\to\infty} \text{mis } I_k = 0}$$

• SE f continua allora: $\forall k$, I_k contiene uno zero di f e

$$\boxed{\lim_{k\to\infty} x_k = \alpha \quad \text{t.c. } f(\alpha)=0}$$

CRITERIO di ARRESTO (necessario: non è possibile costruire tutta una successione in tempo finito)

• di "tipo ASSOLUTO":

dato δ reale positivo...

$$\ldots \boxed{\text{se } \text{mis } I_k < \delta \text{ allora STOP}}$$

1) $\text{mis } I_k = b_k - a_k$ "è calcolabile"

2) disuguaglianza certamente verificata dopo un numero finito di iterazioni...

3) SE f continua:

- $\exists\, \alpha \in I_k$ zero di f
- $|x_k - \alpha| \leq \frac{\text{mis } I_k}{2} < \frac{\delta}{2}$

dunque: si ottiene un'approx di α (utilizzando x_k) con $\boxed{\text{errore assoluto} < \frac{\delta}{2}}$

```
function [z, v, info, k, mis]=bisezione(f_bis, a, b, ...
                          criterio_di_arresto_bis, E_bis, kmax, dialogo)
//
// Uso:
//    [ z,v,info,[k,[mis]] ] = bisezione(f_bis,a,b,...
//                criterio_di_arresto_bis,E_bis,kmax,dialogo)
//
// Metodo di bisezione.
//
// Approssima uno zero della funzione f_bis (che deve essere continua)
// contenuto nell'intervallo [a,b]. La funzione f_bis deve assumere
// valori non nulli e di segno opposto in a e b.
//
// L'iterazione si arresta quando:
//   (*) il valore di f_bis nel punto medio x_m dell'intervallo
//       considerato è zero;
//   (*) la misura dell'intervallo considerato [a(k),b(k)] è minore di
//       E_bis (in tal caso si ha, in teoria, che z approssima uno zero di
//       f_bis con errore assoluto non superiore ad E_bis/2);
//   (*) la funzione f_bis assume valori del medesimo segno negli
//       estremi a e b;
//   (*) dopo kmax iterazioni (valore suggerito: kmax = 55).
//
// z: approssimazione finale (zero di f_bis oppure punto medio
//    dell'ultimo intervallo generato);
// v: valore di f_bis in z;
// info = 0: individuato valore in cui f_bis si annulla;
//      = 1: la misura dell'ultimo intervallo considerato è minore di
//           tol e la f_bis assume valori non nulli e di segno opposto
//           agli estremi di tale intervallo;
//      = 2: f_bis non assume valori non nulli e di segno opposto agli
//           estremi dell'intervallo iniziale [a,b];
//      = 3: superato il numero massimo consentito di iterazioni.
// k: numero di iterazioni effettuate;
// mis: ampiezza dell'ultimo intervallo determinato.
//
// Se info = 2, si pone z = Nan , v = Nan , k = Nan e mis = Nan.
//
info = -1;
k_bis = 0; // contatore delle iterazioni eseguite
tol = E_bis;
if f_bis(a) == 0 then info = 0; z = a; v = 0; k = 0; mis = abs(b-a); end;
if f_bis(b) == 0 then info = 0; z = b; v = 0; k = 0; mis = abs(b-a); end;
if (info ~= 0) & (sign(f_bis(a)) == sign(f_bis(b))) then
        info = 2; v = %nan; z = %nan; k = %nan; mis = %nan; end;
// iterazione
while info == -1,
  x_m = (a + b)/2;
  f_m = f_bis(x_m);
  if f_m == 0 then info = 0; z = x_m; v = 0; k = k_bis; mis = abs(b-a); end;
  if (info ~= 0) & abs(b-a) < tol then info = 1; z = x_m; v = f_m; ...
                                       k = k_bis; mis = abs(b-a); end;
  if k_bis>= kmax then info = 3; z = x_m; v = f_m; ...
                      k = k_bis; mis = abs(b-a);end;
  if info == -1 then k_bis= k_bis+1;
                   if sign(f_m) == sign(f_bis(b)) then b = x_m;
                      else a = x_m; end;
  end;
end;
//
endfunction
```

Es: $f(x) = \cos x$ (continua!)

$I_0 = [1, 2]$

$\delta = 10^{-15}$



Si ottiene (utilizz SCILAB), dopo 50 iterazioni

$$x_{50} = 1,5707\ldots$$

$$\text{mis } I_{50} \approx 8.88\ 10^{-16}$$

Oss: "costo" del calcolo di $x_{50} \approx 50 \times$ (costo 1 it)

$$\text{mis } I_k = \frac{\text{mis } I_0}{2^k} < \delta \iff k > \log_2 \frac{\text{mis } I_0}{\delta}$$

Es: $f(x) = \cos x$, $I_0 = [1,2]$, $\delta = 10^{-16}$

dovrei ottenere la stima di $\left\lceil \log_2 \frac{1}{10^{-16}} \right\rceil = 54$ iterazioni ma...

... dopo 60, 80, 100 it ho $\text{mis } I_{60} = \text{mis } I_{80} = \text{mis } I_{100} \approx 2,22\ 10^{-16}$!!

Per capire il pb, occorre indagare l'aritm del calcolatore.